

# GUI Scripts with KiXforms, Part 2

*Understanding the basic use and structure of KiXforms with VBScript*

by Bob Kelly  
December 2006



Let's get into it by walking through the creation of the simple dialog shown below. This script contains a main form, an image (known as a PictureBox), some text (known as a label) and a button. The button will be instructed to end the script. It is not much, but it is a very good start to seeing just how KiXforms operates and how you can take advantage of it in VBScript.

First, we need to create an object variable that references the KiXforms DLL. The Type Library name is "KiXtart.System" so we begin by setting this object variable:

```
Set System = CreateObject("KiXtart.System")
```

KiXforms is processed from the top to the bottom of the script so we must specify any objects in the order they will be needed. Therefore we will start by establishing the form that will hold our image, text and button. First, set another object variable to create a reference to a form object. Then set the size (height and width) and text (that appears as the window title) for the form:

```
Set Form = System.Form()  
Form.Height = 309  
Form.Width = 281  
Form.Text = "Welcome"
```

Now that we have our form, let's add the text. Doing this is much the same as how we created the form above, but in addition to its size we must also tell KiXforms where the item is to be placed. The location is specified by indicating the number of pixels from the top of the form, and from the left of the form:

```
Set Label = Form.Controls.Label  
Label.Height = 38  
Label.Width = 250  
Label.Top = 199  
Label.Left = 11  
Label.Text = "Welcome to the network, by using this system you " &_  
"agree to the corporate policy for this network."
```



The same properties need to be set for the image, which is referred to as a PictureBox. Additionally, we also specify a *picture* property that identifies the path to the image we wish to display:

```
Set PictureBox = Form.Controls.PictureBox
PictureBox.Height = 175
PictureBox.Width = 250
PictureBox.Top = 12
PictureBox.Left = 11
PictureBox.Picture = "c:\windows\winnt256.bmp"
```

Finally, we will specify the size, location and text for the button. Here, we will also set our first event. When the button is clicked we want to run a function called *ExitScript*. This is set as a string which is to be executed when the event is triggered (in this case the *OnClick* event that happens when someone clicks the button).

```
Set Button = Form.Controls.Button
Button.Height = 23
Button.Width = 75
Button.Top = 240
Button.Left = 100
Button.Text = "OK"
Button.OnClick = "ExitScript()"
```

With our form all laid out, we now need to get the script in a loop where it will look for any events to be triggered. The script has called out all the details of the form so it is ready to show to the user, therefore we set the visible property to true and the form appears on the screen. Without the Do/While loop you see below, the script would end and the user would never catch sight of your nice form. So after setting the form to visible, we instruct the script to execute the events of the form as long as the form remains visible:

```
Form.Visible = "True"
Do While Form.Visible
    Execute(Form.DoEvents)
Loop
```

The DoEvents method runs as long as the script is visible and fires any specified events for the script. In the case of this example, the only event we have handled is the clicking of the button. When the button is clicked, we have it running a function called *ExitScripts* so we provide that function to complete the script:

```
Function ExitScript()
    WScript.Echo "Exiting Script..."
    WScript.Sleep 900
    WScript.Quit
End Function
```

This provides most of the structure you need to get started, and you can get plenty of help from the help file provided along with the KiXforms binary at <http://www.kixforms.org>. In the next (and last) article in this short series on KiXforms, I'll cover some of the more advanced controls that you can't get using a traditional HTA graphical interface and this is where the freely available KiXforms really shines as unique and powerful tool for your scripts. 

*Bob Kelly is president and co-founder of [AdminScriptEditor.com](http://AdminScriptEditor.com), home to an integrated suite of scripting tools and a shared library of scripts and language help. He has authored books on scripting and desktop administration as well as several white papers. Bob also owns and operates [AppDeploy.com](http://AppDeploy.com), where he writes and produces videos on topics related to software deployment. You can contact Bob about "GUI Scripts with KiXforms" at [bkelly@adminscrepteditor.com](mailto:bkelly@adminscrepteditor.com).*